

Per gli amanti del linguaggio macchina e delle utility concettualmente semplici ma molto spinte sul piano implementativo, questo mese il nostro fido Peritore ci mostrerà come modificare ad hoc il boot block del disco di boot per utilizzare i nuovi Amiga indifferentemente in modo PAL o NTSC

TV-Boot

di Giuliano Peritore - Latina

Quasi certamente la cosa che più veniva notata da chi utilizzava uno dei primi Amiga era la banda nera non utilizzata sullo schermo che occupava circa un quinto dell'area visibile. Ciò era dovuto al fatto che negli Stati Uniti a causa dello standard televisivo NTSC il massimo numero di righe visualizzabili era di duecento linee contro le duecentocinquantesi dei sistemi europei PAL. Fino a poco tempo fa l'unico modo per riempire quella fascia di fondo era quello di aumentare le dimensioni verticali della pagina video fino ad occupare tutto il monitor, con gli ovvi incrementi delle dimensioni in RAM delle pagine e l'effetto di schiacciamento delle immagini americane. Tuttavia ora c'è qualcosa di nuovo ed è quello che vedremo in questo articolo.

Le novità

Ultimamente i nuovi Amiga 2000B ed i 500 vengono equipaggiati con il nuovo Fat Agnus, che è in grado di gestire fino ad un megabyte di chip ram. Questo chip, che costituisce il gruppo degli ECS insieme a quello che sarà il nuovo Denise, ha ovviamente anche dei nuovi registri di controllo che consentono di richiamare le nuove funzioni. Il registro che ci interessa fra quelli nuovi è il BEAMCON0 che contiene un bit che ci permette di selezionare il modo PAL od il modo NTSC. La cosa importante è che tale bit cambia effettivamente la frequenza di quadro da 50 Hz (PAL) a 60 Hz (NTSC) tant'è vero che alcuni programmi che si basano sul Vertical Blanking per le temporizzazioni verranno eseguiti più velocemente. Questo cambio di frequenza di quadro porta un grande vantaggio: lo schermo di 200 linee viene allargato fino ad occupare tutto il video esattamente come avviene in un sistema NTSC. L'unico svantaggio è che l'Amiga in modo NTSC non è più miscelabile attraverso Genlock o sistemi analoghi ad altri segnali a frequenza diretta.

Dal punto di vista del software

Osserviamo ora il lato software della faccenda. Il registro BEAMCON0 corri-

sponde alla locazione \$DFF1DC ed il bit NTSC/PAL è il bit 5, settato per il PAL, resettato per l'NTSC. Ovviamente la prima cosa che ci viene in mente è quella di fare una bella MOVE.W #\$0,\$DFF1DC per passare in modo NTSC. Il tutto funziona però c'è l'inconveniente che tutti i parametri di sistema sono settati per il modo PAL e quindi non solo ci ritroveremo senza l'ultima fascia di schermo, ma vedremo anche delle strane cose sul video a causa dei settaggi errati dei vari Display Window Start/Stop, Data Fetch Start/Stop e così via. Si intuisce subito che il modo migliore per realizzare il passaggio è di effettuarlo subito prima di un reset a caldo in modo che tutti i parametri possano essere settati appropriatamente all'avvio del sistema. Ovviamente il meccanismo che più si avvicina alle nostre esigenze è il boot block.

Il programma

Il programma presentato in questo articolo non è altro che un boot block standard leggermente modificato. Quando viene eseguito controlla i pulsanti del mouse. Se nessun pulsante è premuto allora si comporta in modo totalmente normale, altrimenti se è premuto il tasto sinistro setta il bit per il modo PAL e resetta il sistema, se è premuto il tasto destro setta il bit per il modo NTSC e resetta il sistema. Il fatto che venga resettato il sistema può sembrare scomodo ma è in realtà molto utile in quanto estraendo il dischetto da cui si è fatto il boot subito dopo il reset si può inserire qualunque altro disco con la conseguenza di poter quindi fare il boot PAL o NTSC di qualunque programma anche con caricatore nel boot block.

Ma passiamo a descrivere il programma. La prima istruzione controlla se il tasto sinistro del mouse è premuto andando a leggere l'appropriato bit della porta parallela A della CIA A. Se il risultato è positivo salta alla routine apposita. La terza istruzione controlla invece la pressione del tasto destro leggendo il registro POTGOR in cui sono contenuti i segnali provenienti dai piedini delle paddle che possono essere usati anche come porte parallele. Se anche questo controllo è negativo viene eseguito il codice corrispondente al boot block standard che non fa altro quindi che cercare il modulo residente del

```

;TV-Boot      Public Domain by Giuliano C. Peritore
;              Via Amaseno, 6
;              04100 Latina - ITALY
;Durante il Boot premere il tasto sinistro per selezionare
;il modo PAL, il tasto destro per selezionare il modo NTSC,
;oppure nessun tasto.
;Quando il computer inizia a resettarsi di nuovo lasciare
;il tasto premuto

BEAMCON0      EQU      $DFF1DC
CIAAPRA       EQU      $BFE001
POTGOR        EQU      $DFF016
FindResident  EQU      -96
SuperState    EQU      -150

TVBoot:
    dc.l      $444f5300      ;DOS\0
    dc.l      $c6a81c8c      ;Checksum
    dc.l      $00000370      ;Root Pointer
    btst      #6,CIAAPRA     ;Left Button Pressed ?
    beq.s     PAL            ;Yes
    btst      #10,POTGOR     ;Right Button Pressed ?
    beq.s     NTSC          ;Yes

    lea      DosName(pc),a1
    jsr      FindResident(a6)
    tst.l     d0
    beq.s     NotFound
    move.l   d0,a0
    move.l   $16(a0),a0
    moveq    #$0,d0
    rts

NotFound:
    moveq    #$ff,d0
    rts

PAL:
    moveq    #$20,d2        ;Set PAL bit

    bra.s    DoHack

NTSC:
    moveq    #0,d2         ;Clear PAL bit

DoHack:
    move.l   4,a6
    jsr      SuperState(a6)
    lea      BEAMCON0-2,a3
    move.l   2,a6           ;Reset Routine
    lea      $FF4794,a5    ;Kickstart Pointer

;Ecco il pezzo di kickstart 1.3 situato a $FF4794
;00FF4794 2682          move.l   D2,(A3)
;00FF4796 4ED6          jmp      (A6)

    reset
    jmp      (a5)          ;DoIt !

DosName:
    dc.b     'dos.library',0
    dc.b     'TV-Boot - by Giuliano C. Peritore 1990'
    dc.b     ' - This program is Public Domain'
    dcb.b    1024,0

```

DOS e agire di conseguenza. Se abbi-amo invece premuto uno dei due tasti verrà messo in D2 il valore che dovrà essere messo in BEAMCON0 e poi verrà eseguita la routine DoHack. In questa routine viene eseguito un piccolo truccetto. Il problema era quello di resettare l'Amiga, settare appropriatamente il bit PAL/NTSC dopo il RESET e saltare all'inizio del Kickstart. È importante settare il bit dopo il RESET in quanto altrimenti verrebbe settato in base allo stato del jumper PAL/NTSC presente sulla piastra madre dell'Amiga. L'istruzione RESET causa però uno svantaggio. Quando l'Amiga parte a freddo, o durante un RESET, la RAM

viene totalmente disabilitata. Va da sé che non sarebbe possibile fare praticamente nulla dopo l'istruzione RESET. Se però cerchiamo un pezzo del Kickstart che si addice ai nostri scopi possiamo saltarci facendo in modo che venga eseguito dopo il RESET in quanto si trova in ROM senza quindi doverci preoccupare della RAM. Ho trovato un pezzo di Kickstart adatto a \$FF4794. Ricordatevi che tale indirizzo vale per il Kickstart 1.3 solamente, ma del resto se avete il Fat Agnus avete anche la versione 1.3. Per raggiungere i nostri scopi basta semplicemente mettere il valore del registro in D2, l'indirizzo del registro -2 in A3 e l'indirizzo della loca-

zione 2 in A6. Tutto questo perché il 68000 legge la word di programma successiva dalla RAM prima di terminare l'esecuzione dell'istruzione precedente e quindi la Jmp (A5) viene letta prima del RESET e può quindi essere eseguita dopo. Non ci importa più della RAM perché ormai il controllo verrà trasferito in ROM (a \$FF4794) che setterà il bit e salterà alla routine il cui indirizzo è contenuto in A6.

Dopo aver compilato il programma basta caricarlo con un monitor (ad esempio il Rossimon), ricalcolare il checksum per sicurezza e scrivere il boot block su disco.

Considerazioni

Sebbene la visualizzazione in NTSC non dia alcun fastidio, ed anzi ci riporta alla realtà delle foto che siamo abituati a vedere sui manuali e che a noi, poveri (perché poi) utenti PAL sembrano paurosamente allungate, potremmo avere qualche problema. Infatti mentre un programma NTSC non avrà quasi mai problemi a girare su macchine PAL in quanto al massimo lascerebbe inutilizzata una fascia di schermo, un programma espressamente progettato per gli Amiga PAL potrebbe avere seri problemi a girare in un ambiente NTSC. Inoltre è abbastanza diffusa la pratica di proteggere i programmi in modo che le versioni PAL non possano girare sulle macchine NTSC e viceversa. Un classico esempio di programma che potrebbe sorprenderci è il DPaint che se viene caricato in modo PAL su un Amiga bootato in NTSC non ci consente di sfruttare le ultime cinquantasei righe della pagina 320x256 in quanto il mouse (settato per un video NTSC) non scende più giù della duecentesima riga. Un altro programma che presenta un inconveniente è il Soundtracker con il quale la selezione della modalità NTSC provoca un aumento di velocità di sei quinti (60/50Hz) nell'esecuzione della musica.

Rimangono tuttavia tutti i vantaggi che tale tecnica può offrire a chi sviluppa software destinato sia alle macchine europee che a quelle americane che può quindi provare il suo software senza problemi in entrambi gli ambienti senza dover aspettare che a causa dell'errore presente nella Graphics.Library 1.2 la propria macchina PAL facesse il boot in pseudo NTSC!!! Arrivederci.

